
Forecasting in STATA: Tools and Tricks 

 

Introduction 

This manual is intended to be a reference guide for time‐series forecasting in STATA.  It will be updated 
periodically during the semester, and will be available on the course website. 

Working with variables in STATA 

In the Data Editor, you can see that variables are recorded by STATA in spreadsheet format. Each rows is 
an observation, each column is a different variable. An easy way to get data into STATA is by cutting‐
and‐pasting into the Data Editor.  

When variables are pasted into STATA, they are given the default names “var1”, “var2”, etc.  You should 
rename them so you can keep track of what they are. The command to rename “var1” as “gdp” is: 

. rename var1 gdp 

New variables can be created by using the generate command. For example, to take the log of the 
variable gdp: 

. generate y=ln(gdp) 

 

Dates and Time 

For time‐series analysis, dates and times are critical. You need to have one variable which records the 
time index. We describe how to create this series. 

Annual Data 

For annual data it is convenient if the time index is the year number (e.g. 2010). Suppose your first 
observation is the year 1947. You can generate the time index by the commands: 

. generate t=1947+_n-1 

. tsset t, annual 

The variable “_n” is the natural index of the observation, starting at 1 and running to the number of 
observations n. The  generate command creates a variable “t” which adds 1947 to “_n”, and then 
subtracts 1, so it is a series with entries “1947”, “1948”, “1949”, etc.  The tsset command declares the 
variable “t” to be the time index. The option “annual” is not necessary, but tells STATA that the time 
index is measured at the annual frequency. 



Quarterly Data 

STATA stores the time index as an integer series. It uses the convention that the first quarter of 1960 is 
0. The second quarter of 1960 is 1, the first quarter of 1961 is 4, etc. Dates before 1960 are negative 
integers, so that the fourth quarter of 1959 is ‐1, the third is ‐2, etc.  

When formatted as a date, STATA displays quarterly time periods as “1957q2”, meaning the second 
quarter of 1957. (Even though STATA stores the number “‐11”, the eleventh quarter before 1960q1.) 
STATA uses the formula “tq(1957q2)” to translate the formatted date “1957q2” to the numerical index 
“‐11”.   

Suppose that your first observation is the third quarter of 1947. You can generate a time index for the 
data set by the commands 

. generate t=tq(1947q3)+_n-1 

. format t %tq 

. tsset t 

The generate command creates a variable “t” with integer entries, normalized so that 0 occurs in 
1060q1. The format command formats the variable “t” using the time‐series quarterly format. The “tq” 
refers to “time‐series quarterly”.  The tsset command declares that the variable “t” is the time index. 
You could have alternatively typed 

. tsset t, quarterly 

to tell STATA that it is a quarterly series, but it is not necessary as “t” has already been formatted as 
quarterly. Now, when you look at the variable “t” you will see it displayed in year‐quarter format. 

 

Monthly Data 

Monthly data is similar, but with “m” replacing “q”. STATA stores the time index with the convention 
that 1960m1 is 0. To generate a monthly index starting in the second month of 1962, use the commands 

. generate t=tm(1962m2)+_n-1 

. format t %tm 

. tsset t 

 

 

 



Weekly Data 

Weekly data is similar, with “w” instead of “q” and “m”, and the base period is 1960w1. For a series 
starting in the 7th week of 1973, use the commands 

. generate t=tw(1973w7)+_n-1 

. format t %tw 

. tsset t 

 

Daily Data 

Daily data is stored by dates.  For example, “01jan1960” is Jan 1, 1960, which is the base period. To 
generate a daily time index staring on April 18, 1962, use the commands 

. generate t=td(18apr1962)+_n-1 

. format t %td 

. tsset t 

 

Pasting a Data Table into STATA 

Some quarterly and monthly data are available as tables where each row is a year and the columns are 
different quarters or months. If you paste this table into STATA, it will treat each column (each month) 
as a separate variable. You can use STATA to rearrange the data into a single column, but you have to do 
this for one variable at a time. 

I will describe this for monthly data, but the steps are the same for quarterly. 

After you have pasted the data into STATA, suppose that there are 13 columns, where one is the year 
number (e.g. 1958) and the other 12 are the values for the variable itself. Rename the year number as 
“year”, and leave the other 12 variables listed as “var2” etc. Then use the reshape command 

. reshape long var, i(year) j(month) 

Now, the data editor should show three variables: “year”, “month” and “var”. STATA has resorted the 
observations into a single column. You can drop the year and month variables, create a monthly time 
index, and rename “var” to be more descriptive.  

In the reshape command listed above, STATA takes the variables which start with “var” and strips off the 
trailing numbers and puts them in the new variable “month”. It uses the existing variable “year” to 
group observations. 



Data Organized in Rows 

Some data sets are posted in rows. Each row is a different variable, and each column is a different time 
period. If you cut and paste a row of data into STATA, it will interpret the data as a single observation 
with many variables.  

One method to solve this problem is with Excel. Copy the row of data, open a clean Excel Worksheet, 
and use the Paste Special Command. (Right click, then “Paste Special”.) Check the “Transpose” option, 
and “OK”. This will paste the data into a column. You can then copy and paste the column of data into 
the STATA Data Editor. 

 

Cleaning Data Pasted into STATA 

Many data sets posted on the web are not immediately useful for numerical analysis, as they are not in 
calendar order, or have extra characters, columns, or rows. Before attempting analysis, be sure to 
visually inspect the data to be sure that you do not have nonsense. 

Examples 

• Data at the end of the sample might be preliminary estimates, and be footnoted or marked to 
indicate that they are preliminary. You can use these observations, but you need to delete all 
characters and non‐numerical components.  Typically, you will need to do this by hand, entry‐
by‐entry. 

• Seasonal data may be reported using an extra entry for annual values. So monthly data might be 
reported as 13 numbers, one for each month plus 1 for the annual. You need to delete the 
annual variable. To do this, you can typically use the drop command. For example, if these 
entries are marked “Annual”, and you have pasted this label into “var2”, then 

. drop if var2==”Annual” 

This deletes all observations for which the variable “var2” equals “Annual”. Notices that this 
command uses a double equality “==”. This is common in programming. The single equality “=” 
is used for assignment (definition), and the double equality “==” is used for testing. 

Time‐Series Plots 

The tsline command generates time‐series plots. To make plots of the variable “gdp”, or the variables 
“men” and “women” 

. tsline gdp 

. tsline men women 



Time‐series operators  

For a time‐series y 

L.  lag y(t‐1) 

  Example: L.y 

L2.  2‐period lag y(t‐2) 

  Example: L2.y 

F.  lead y(t+1) 

  Example: F.y 

F.  2‐period lead y(t+2) 

  Example: F2.y 

D.  difference y(t)‐y(t‐1) 

  Example: D.y 

D2.  double difference (y(t)‐y(t‐1))‐ (y(t‐1)‐y(t‐2)) 

  Example: D2.y 

S.  seasonal difference y(t)‐y(t‐s), where s is the seasonal frequency (e.g., s=4 for quarterly) 

  Example: S.y 

S2.  2‐period seasonal difference y(t)‐y(t‐2s) 

  Example: S2.y 

 



Regression Estimation 

To estimate a linear regression of the variable y on the variables x and z, use the regress command  

. regress y x z 

The regress command reports many statistics. In particular, 

• The number of observations is at the top of the small table on the right 

• The sum of squared residuals is in the first column of the table on the left (under SS), in the row 
marked “Residual”.  

• The least‐squares estimate of the error variance is in the same table, under “MS” and in the row 
“Residual”. The estimate of the error standard deviation is its square root, and is in the right 
table, reported as “Root MSE”. 

• The coefficient estimates are repoted in the bottom table, under “Coef”. 

• Standard errors for the coefficients are to the right of the estimates, under “Std. Err.” 

In some time‐series cases (most importantly, trend estimation and h‐step‐ahead forecasts), the least‐
squares standard errors are inappropriate. To get appropriate standard errors, use the newey command 
instead of regress.  

. newey y x z, lag(k) 

Here, “k” is an integer, meaning number of periods, which you select. It is the number of adjacent 
periods to smooth over to adjust the standard errors. STATA does not select k automatically, and it is 
beyond the scope of this course to estimate k from the sample, so you will have to specify its value.  I 
suggest the following. In h‐step‐ahead forecasting, set k=h. In trend estimation, set k=4 for quarterly and 
k=12 for monthly data. 

 

Intercept‐Only Model 

The simplest regression model is intercept‐only, y=b0+e. This can be estimated by the regress or newey 
command 

. regress y 

. newey y, lag(k) 

The estimated intercept is the sample mean of “y”. While this could have been calculated using other 
methods, such as the summarize command, using the regress/newey command is useful as then 
afterwards you can use postestimation commands, including predict. 



Regression Fit and Residuals 

To calculate predicted values, use the predict command after the regress or newey command 

. predict p 

This creates a variable “p” of the fitted values  x’beta. 

To calculate least‐squares residuals, after the regress or newey command 

. predict e, residuals 

This creates a variable “e” of the in‐sample residuals  y‐x’beta. 

You can then plot the fit versus actual values, and a residual time‐series 

. tsline y p 

. tsline e 

The first plot is a graph of the variables y and p, assuming that y is the dependent variable, and p are the 
fitted values. The second plot is a graph of the residuals against time.  

 

Dummy Variables 

Indicator variables, known as dummy variables, can be created using generate. One purpose is to create 
sub‐periods and regimes.  

For example, to create a dummy variable equaling “0” for observations before 1984, and equaling “1” 
for monthly observations starting in 1984 

. generate d=(t>=tm(1984m1)) 

In this example, the time index is “t”. The command tm(1984m1)converts the date format 1984m1 
into an integer value. The new variable is “d”, and equals “0” for observations up to 1983m12, and 
equals “1” for observations starting in 1984m1.  

To create a dummy variable equaling “1” for quarterly observations between 1990q1 and 1998q4, and 
“0” otherwise, (and the time index is “t”) use 

. generate d=(t>=tq(1990q1))*(t<=tq(1998q4)) 

This command essentially generated two dummy variables and then multiplied them to create the 
variable “d”.  

 



Changing Intercept Model 

We can allow the intercept of a model to change at a known time period we simply add a dummy 
variable to the regression. For example, if “t” is the time index, the data are monthly and we want a 
change in mean starting in the 7th month of 1987, 

. generate d=(t>=tm(1987m7)) 

. regress y d 

The generate command created a dummy variable for the second time period. The regress command 
estimated an intercept‐only model allowing a switch in the intercept in July 1987.  

The estimated “constant” is the intercept before July 1987. The coefficient on “d” is the change in the 
intercept.  

 

Time Trend Model 

To estimate a regression on a time trend only, use regress or newey with the time index as a regressor. 
If the time index is “t” 

. regress y t 

 

Trends with Changing Slope 

Here is how to create a trend which changes slope at a specific date  (for concreteness 1984m1). Use the 
generate command to create a dummy for the period starting at 1984m1, and then interact it with a 
trend normalized to be zero at 1984m1: 

. generate d=(t>=tm(1984m1)) 

. generate ts=d*(t-tm(1984m1)) 

The new variable “ts” is zero before 1984, and then is a linear trend after that.   

Then regress the variable of interest on “t” and “ts”: 

. regress t ts 

The coefficient on “t” is the trend before 1984. The coefficient on “ts” is the change in the trend. 

If you want there to be a jump as well as a change in slope at 1984m1, then include the dummy “d” 

. regress t d ts 



Expanding the Dataset Before Forecasting 

When you have a set of time‐series observations, STATA typically records the dates as running from the 
first until the last observation. You can check this by looking at the data in the Data Editor. But to 
forecast a date out‐of‐sample, these dates need to be in the data set. This requires expanding the 
dataset to include these dates. This is done by the tsappend command. There are two formats 

. tsappend, add(12) 

This command adds 12 dates to the end of the sample. If the current final observation is 2009m12, the 
command adds 2010m01 through 2010m12. If you look at the data using the Data Editor, you will see 
that the time index has new entries, through 2010m12, but the other variables are missing. Missing 
values are indicated by a period “.”.   

The other format which accomplishes the same task is 

. tsappend, last (2010m12) tsfmt(tm) 

This command adds observations so that the last observation is 2010m12, and that the formatting is 
monthly.  For quarterly data, to add observations up to 2010q4 the command is 

. tsappend, last (2010q4) tsfmt(tq) 

 

Point Forecasting Out‐of‐Sample 

The predict command can be used for point forecasting, so long as the regressors are available. The 
dataset first needs to be expanded as previously described, and the regression coefficients estimated 
using either the regress or newey commands.  

The command 

. predict p 

will create a series “p” of predicted values, both in‐sample and out‐of‐sample. To restrict the predicted 
values to be in‐sample, use 

. predict p 

To restrict the predicted values to in‐sample observations (for quarterly data with time index “t” and the 
last in‐sample observation 2009m12) 

. predict p if t<=tm(2009m12) 

To restrict the predicted values to out‐of‐sample (for monthly data with the last in‐sample 2009m12) 

. predict yp if t>tm(2009m12) 



If the observations, in‐sample predictions, and out‐of‐sample predictions are y, p, and yp, they can be 
plotted together, but as three distinct elements, as 

. tsline y p yp 

. tsline y p yp if t>tm(2000m12) 

The second command restricts the plot to observations after 2000, which is useful if you wish to focus in 
on the forecast period (the example is for quarterly data). 

 

Normal Forecast Intervals 

To make an interval forecast based on the normal approximation, you need what are called the 
“standard deviation of the forecast”, which is an estimate of the standard deviation of the forecast 
error. These are computed using the predict command. You first need to estimate the forecast and save 
the forecast. Suppose you are forecasting the monthly variable “y” given the regressors “x” and “z”, the 
in‐sample ends in 2009m12 and we make the following commands 

. regress y x z 

. predict p if t<=tm(2009m12) 

. predict yp if t>tm(2009m12) 

Then you add 

. predict s if t>tm(2009m12), stdf 

This creates a variable “s” for the forecast period whose entries are the standard deviation of the 
forecast. Now you multiply this by a standard normal quantile and add to the point forecast 

. generate yp1=yp-1.645*stdf 

. generate yp2=yp+1.645*stdf 

These commands create two series for the forecast period, which equal the endpoints of a forecast 
interval with 90% coverage. (‐1.645 and 1.645 are the 5% and 95% quantiles of the normal distribution).  

 

Empirical Forecast Intervals 

To make an interval forecast, you need to estimate the quantiles of the residuals of the forecast 
equation. To do so, you first need to estimate the forecast and save the forecast. Suppose you are 
forecasting the monthly variable “y” given the regressors “x” and “z”, the in‐sample ends in 2009m12 
and we make the following commands 



. regress y x z 

. predict p if t<=tm(2009m12) 

. predict yp if t>tm(2009m12) 

. predict e, residuals 

Now we want to calculate the 25% and 75% quantiles of the residuals “e”. This can be accomplished 
using what is called quantile regression with just an intercept. The STATA command is qreg. The format 
is similar to regress, but you have to tell STATA the quantile you want to estimate. 

. qreg e, quantile(.25) 

This command computes the 25% quantile regression of “e” on an intercept (as no regressors are 
specified). The “Coef.” Reported in the table is the .25 quantile of “e”. Now you can compute the out‐of‐
sample values, and add them to the point forecast “yp” to create the lower part of the forecast interval 

. predict q1 if t>tm(2009m12) 

. generate yp1=yp+q1 

The predict command uses the last estimation command – in this case qreg – to compute the forecast. 
In this case it is computing the out‐of‐sample .25 quantile of “e”.  

You can repeat this for the upper forecast interval endpoint. 

. qreg e, quantile(.75) 

. predict q2 if t>tm(2009m12) 

. generate yp2=yp+q2 

The variables “yp1” and “yp2” are the out‐of‐sample forecast interval endpoints for “y”. You can plot the 
data together with the out‐of‐sample point and interval forecasts, e.g. 

. tsline y yp yp1 yp2 if t>tm(2000m12) 

For a fan chart, you repeat this for multiple quantiles. 

 

Conditional Forecast Intervals 

The qreg command makes it easy to compute the forecast interval endpoints conditional on regressors. 
This is a quite advanced technique, so I do not recommend it without care. But this is how it can be 
done. As in the previous section, suppose you are forecasting “y” given “x” and “z”, have forecast 
residuals “e”, and out‐of‐sample point forecast “yp”. Now you want out‐of‐sample conditional quantiles 



of “e” given some regressors. Suppose that you think that “x” has predictive power for the quantiles of 
“e”. You can use the commands for the .25 quantile 

. qreg e x, quantile(.25) 

. predict q1 if t>tm(2009m12) 

. generate yp1=yp+q1 

and similarly for the .75 quantile.  

This method models the quantiles of “e” as functions of “x”. This can be useful when the spread 
(variance) of the distribution changes over time.   


